
BASICODE

1. Das Grundprinzip

BASICODE wurde von niederlandischen Computerspezialisten um 1979 entwik-
kelt, wird in den Niederlanden regelmafig im Rundfunk verwendet und wurde
fortlaufend verbessert. BASICODE ermdglicht, daBR unabhdngig vom verwendeten
Rechnertyp Programme ausgetauscht und im Rundfunk Ubertragen werden kénnen.
Heute existiert 1in den Niederlanden eine gewinnfrei arbeitende Stiftung
unter dem Vorsitz von Klaas Robert, der auch entscheidender Initiator des
BASICODE ist. Von dieser Stiftung hat der Rundfunk der DDR die Rechte zur
Anwendung erworben. Mit der Entwicklung der Computertechnik wurde auch
BASICODE verbessert. Aus diesem Grunde existiert jetzt die dritte Version
BASICODE-3, welche wir verwenden. BASICODE-4 ist in Vorbereitung und be-
sitzt dann zusatzliche Méglichkeiten. (vgl. Abschnitt 12)

BASICODE kann als eine spezielle Variante von BASIC angesehen werden. In
reichlich 50 Befehlen ist BASICODE identisch mit BASIC (siehe Anhang 1).
Weitere 30 BASIC-Befehle werden indirekt (iber spezielle GOSUB-Routinen
realisiert (Anhang 2). Dies sind Befehle, die auf verschiedenen Rechnern
unterschiedlich realisiert oder z.T. nicht vorhanden sind. So existiert nur
bei wenigen Rechnern die Méglichkeit LOCATE (Spalte, Zeile), welche den
Cursor an eine bestimmte Bildschirm-Position setzt. Mit BASICODE existiert
hierzu GOSUB 110, wobei vorher auf die Variablen HO und VE die horizontale
bzw. vertikale Koordinate zu lbergeben ist. Da nicht alle Rechner die glei-
che Anzahl Zeilen und Spalten besitzen, erfolgt durch BASICODE auch eine
Uberpriifung auf die Zuldssigkeit der Werte.

In BASICODE-3 existieren weiter einheitliche Grafik- und Ton-Routinen. Sie
erméglichen, dal sich Bild und Ton automatisch den Gegebenheiten des Rech-
ners anpassen. Eine Grafik erscheint so bei allen Rechnern in gleicher Form
und am gleichen Ort auf dem Bildschirm. Lediglich die Qualitat des Bildes
hangt von den Mdglichkeiten des Rechners ab. Hdéchste Qualitat kann so bei
IBM-kompatiblen Rechnern mit Hercules- oder EGA-Karte erreicht werden. Der
KC 85/3 schneidet auch recht gut ab. Aber selbst der KC 87, der eigentlich
keine echte Grafik besitzt, 1liefert noch brauchbare Bilder. Beim Ton ist
fir praktisch alle Rechner eine brauchbare Melodie in richtiger Tonhdhe
erreichbar.

Doch damit sind noch keineswegs die Vorteile von BASICODE erschopft. Ein
besonders wichtiger ist dadurch gegeben, dall alle BASICODE-Programme ein-
heitlich auf Kassette aufgezeichnet werden. Eine Kassette, die vom BASICODE
von irgendeinem Rechner, z.B. einem Spectrum, aufgezeichnet wurde, ist auf
jedem anderen Rechner wieder lesbar. Dem Austausch von BASICODE-Programmen
sind so keine Grenzen gesetzt. Deshalb brauchen wir im Rundfunk ein
BASICODE-Programm auch nur in dieser einen Version zu senden. Stellen Sie
sich nur einmal vor, daB auf diese Weise alle Rechnerfreunde problemlos
miteinander in Verbindung treten kdénnen. Die Vorteile dieses Verfahrens
sind also lberhaupt nicht abzusehen.

Weiter sei erwahnt, dall viele Bascoder so geschrieben sind, daR die Pro-
gramme zuweilen durchaus schneller als normale BASIC-Programme ablaufen.
Natirlich gibt es auch Nachteile. Die sind jedoch vergleichsweise gering
und betreffen folgende Punkte:

Erstens ist ein zusatzliches Programm, der "Bascoder" (siehe unten), vorher
zu laden. Dieses Programm 1ist rechnerspezifisch und muB damit fir jeden
Rechner zur Verfligung stehen. Dank der niederlandischen Spezialisten stehen
uns die Bascoder fir die meisten Importrechner bereits zur Verfligung. Fir
die DDR-Rechner konnten inzwischen von DDR-Computerfreunden ebenfalls Bas-
coder entwickelt werden. Weitere werden gewill noch entstehen. Die Bascoder

fur die wichtigsten Rechner stehen somit, von geringen Unkosten abgesehen,
unentgeltlich zur Verfigung.

Zweitens ist das Programmieren in BASICODE etwas ungewohnt und es missen
mehrere Vorschriften streng eingehalten werden, denn der Rechner selbst
kann auch mit dem Bascoder nicht feststellen, ob unerlaubte Befehle verwen-
det werden, die zwar auf diesem Rechner, jedoch nicht auf anderen, laufen.
Drittens bendtigen einige Rechner eine geringe zusatzliche Hardware.

2. Betrieb in BASICODE

Fir jeden Rechner existiert ein besonderer Bascoder, der jeweils von einem
anderen Experten entwickelt wurde. Hierdurch und durch die Spezifika der
Hardware ist es nicht moéglich, eine allgemein giiltige Anleitung fir den Um-
gang mit BASICODE zu geben. Dennoch gibt es weitgehend glltige Grundsatze.
Sie seien im Folgenden dargestellt.

Bei den normalen BASICs existieren drei Schichten des Betriebes:

1. Das Betriebssystem.
Bei den KC-Rechnern CAOS bzw. 0S. Hier kénnen z.B. Maschinenprogramme
geladen und abgearbeitet werden.

2. Der BASIC-Betrieb (Direkt-Mode).
Er bestatigt Eingaben wie CLS, LIST usw. nach der Abarbeit mit OK oder
bei anderen Rechnern mit READY.

3. Die Abarbeitung eines BASIC-Programms.
Sie wird durch den Befehl RUN eingeleitet.

Beim BASICODE wird in der Regel in 2. der Bascoder geladen. Dies ist ein
spezielles BASIC-Programm, welches meist Maschinenroutinen enthalt und es
erméglicht, BASICODE-Programme unter Zuhilfenahme des existierenden BASIC-
Interpreters abzuarbeiten. Der Bascoder lagert sich also nach dem Punkt 2.
an und es entstehen in etwa vier Schichten:

1. Betriebssystem
2. BASIC-Betrieb
3. Bascoder

4. BASICODE-Mode

Erst, nachdem der Bascoder (und meist auch das BASICODE-Pogramm) geladen
ist, kann der BASICODE-Mode erreicht werden. Hier kann die Abarbeitung
eines ebenfalls geladenen BASCICODE-Programms beginnen. Sein Start erfolgt
mit

RUN

Der BASICODE-Mode wird also mittelbar (iber den Bascoder bewirkt und ist
ein besonderer Zustand TIhres Rechners. Dieser Zustand ist zur Beendigung
des BASICODE-Mode folglich auch wieder zu verlassen. Aus diesem Grunde
enden BASICODE-Programme auch nicht wie Ubliche BASIC-Programme, z.B. durch
ein END bzw. STOP im Programm, oder ohne besondere Hinweise, indem das Pro-
gramm eben einfach aufhért. Ein BASICODE-Programm muB mit GOTO 950 enden.
Diese Grundsatze wollen wir mit unserem ersten Programm "HYDRA" demonstrie-
ren: Es beginnt mit der Zeile 1000. Sie entspricht der CLEAR-Anweisung im
normalen BASIC. A=100 bedeutet, dal ein Stringraum der GrofRe hundert Byte
reserviert wird. GOTO 20 fidhrt zum Bascoder und initialisiert den BASICODE-

Mode. Dabei werden die fir den Rechner notwendigen Spezifika ausgelést, die
Variablen geldéscht und weitere notwendige Aktionen bewirkt. Vom Bascoder
erfolgt dann ein Ricksprung zur Zeile 1010 und die Abarbeitung des
BASICODE-Programms beginnt. Nach seiner Abarbeitung - sie ist bei diesem
speziellen Programm bei Zeile 1030 erreicht - erfolgt der Sprung mit
GOTO 950 in den Bascoder. Er hebt dort den BASICODE-Mode auf und fuhrt den
Rechner in den normalen BASIC-Zustand zurick. Es erscheint wieder das 0K
bzw. das READY. Aus diesen Ausfihrungen wird deutlich, warum innerhalb ei-
nes BASICODE-Programms

RUN, END und STOP

verboten sind (verbotene Befehle fallt Tabelle 3 zusammen). Weiter mull immer
je eine Zeile 1000 und 1010 existieren. In der Zeile 1000 sollte der not-
wendige Stringraum lber die Variable A festgelegt werden und danach erfolgt
ein Sprung zur Zeile 20. Dieser Aufbau der Zeile 20 ist notwendig. Vor-
schriftsmalfig sollte die Zeile mit einem REM und dem Namen des Programms
fortgesetzt werden.

1000 A=100: GOTO 20: REM ### HYDRA ###

1010 PRINT "Hydra-Problem": PRINT

1020 PRINT "Eingabe einer Zahl: 0 fuer Ende ";
1025 INPUT Z

1030 IF Z=0 THEN 950: REM Ende

1040 IF (Z<>INT(Z)) OR (Z<2) THEN PRINT"Fehler": GOTO 1020
1050 Y=Z/2: IF Y=INT(Y) THEN Z=Y: GOTO 1070

1060 Z=Z+Z+Z+1

1070 PRINT Z;: IF Z>1 THEN 1040

1080 PRINT: GOTO 1020

1090 REM ----Programm-Ende------

30000 REM Das Hydra-Problem stammt von McCarthy.
30010 REM Es wird vermutet, dass alle ganzzahligen
30020 REM Eingaben zu 1 und damit zum Ende fuehren.
30030 REM Ein Beweis steht bis heute aus.

30040 REM Infolge der endlichen Arithmetik kann

30050 REM ein Rechner-Programm nur fuer einige Zahlen
30060 REM diese Hypothese unterstuetzen.

30070 REM Beobachten Sie das Zu- und Abnehmen

30080 REM der Zahlenwerte und versuchen Sie

30090 REM das Programm zu verstehen.

32000 REM ------cmmmmmmm i -

32010 REM H. Voelz; 9.5.89; 24.8.89 fuer Rundfunk
32020 REM XT-compatibler Rechner

Die Funktion des Programms besteht darin, eine Zahl dann durch zwei zu
teilen, wenn sie geradzahlig ist, anderenfalls sie mit drei zu multiplizie-
ren und anschlieBend eins zu addieren (Zeilen 1040 bis 1060). Dies ist
fortlaufend zu wiederholen, bis 1 erreicht wird. Andere Bemerkungen stehen
in den Kommentarzeilen. Ihre Zeilennummern sind fir BASICODE genau festge-
legt (Anhang 4). Ab 30000 stehen Erklarungen zum Programm und ab 32000
formale Fakten wie Autor, Datum, Versionsnummer, verwendeter Rechner usw.

3. Einfache Befehle im Text-Mode

Nach der ersten Einfihrung in BASICODE werden nun schrittweise weitere
GOSUB-Routinen erklart. Die nachsten vier Programme betreffen den Text-
Mode. Der zweite Mode - der Grafik-Mode - wird im Abschnitt 5 behandelt.

1000 A=100: GOTO 20: REM ### DATEN ###
1010 PRINT "Bildschirmdaten"
1020 PRINT"Text: Spalten * Zeilen = ";HO;"*";VE

1030 PRINT"Pixel: X * Y ="; HG;"*";VG
1040 PRINT"Taste betaetigen": GOSUB 210: GOTO 950
1050 REM ----Programm-Ende------

30000 REM Anzeige der vorhandenen Bildschirmwerte.
30010 REM Es existieren zwei Modi:

30020 REM Zeilen/Spalten fuer Text

30030 REM Pixel fuer Grafik

32000 REM -------cmmmm i

32010 REM H. Voelz; 10.5.89; 24.8.89

32020 REM XT-compatibler Rechner

Das zuvor stehende Programm "DATEN" demonstriert drei Fakten:

1) Anhalten des Programms, bis eine Taste betatigt wird,
2) Anzeige der spezifischen Parameter des jeweiligen Rechners,
3) den Umgang mit BASICODE-spezifischen Variablen.

Mit GOTO 20 werden automatisch vier spezielle Variablen mit Daten belegt,
die fur den angewendeten Rechner typisch sind:

HO enthdlt die Anzahl der mdéglichen Spalten -1,
VE enthalt die Anzahl der mdéglichen Zeilen -1,
HG enthdlt die Anzahl der méglichen Pixel in x-Richtung,
VG enthalt die Anzahl der mdéglichen Pixel in y-Richtung.

Diese Werte werden in den Zeilen 1020 und 1030 angezeigt. Damit ist das
Programm eigentlich bereits zu Ende. Jedoch, wenn darauf GOTO 950 unmittel-
bar folgen wiirde, ginge der Rechner sofort in den BASIC-Mode iiber und alles
ware geldscht. Deshalb muB man die Anzeige so lange sichtbar halten, bis
eine Taste betdtigt wird. Im normalen BASIC geschieht dies oft mit: INPUT A
oder INPUT"";A. Bei beiden erscheint aber zumindest der Cursor, eventuell
auch noch ein Fragezeichen. BASICODE besitzt einen besseren Befehl mit
GOSUB 210. Das Programm wird es Ihnen zeigen.

Neben den oben genannten vier Variablen (HO, VE, HG, VG) und dem A in Zeile
1000 verwendet BASICODE weitere spezifische Variablen, auf die beim Pro-
grammieren achtzugeben ist. Sie werden noch erlautert und sind im Anhang in
Tabelle 5 zusammengestellt. Dariiber hinaus sind auch einige Variablen -
ahnlich wie im normalen BASIC - verboten. Sie faRt Tabelle 6 zusammen.

Das nachste Programm "ZUTEXT" erzeugt Zufallstext und verteilt ihn unregel-
malig lber den Bildschirm. Hierzu verwenden viele BASIC-Dialekte den Befehl
RND. Da er aber nicht in allen Rechnern existiert, wird stattdessen in
BASICODE die Anweisung GOSUB 260 benutzt. Mit ihr wird in die Variable RV
ein Zufallswert gemaB 0<=RV<l abgelegt. Ein "RANDOMIZE" ist ebenfalls bei
der Initialisierung lber GOTO 20 vorhanden. Dadurch ist die Pseudozufalls-
folge in ihrem Start bei RUN unbestimmt.

Die zufallige Anordnung auf dem Bildschirm erfolgt mittels eines Befehls,
der PRINT AT ahnelt und durch GOSUB 110 realisiert wird. Die Zeilen- und
Spaltenpositionen werden hierbei mittels der Variablen HO und VE libergeben.
Dies sind genau jene Variablen, die nach GOTO 20 die Spalten- bzw. Zeilen-
zahl enthalten. Deshalb wurden diese Werte in 1010 auf die Variablen HT und
VT zur Normierung der Zufallszahlen lbergeben. So kann erreicht werden, dal
bei jedem Rechner der Bildschirm gleichmaBfig beschrieben wird. In dem Pro-
gramm fallt das GOSUB 100 in Zeile 1080 auf. Es entspricht im Wesentlichen
dem Léschen des Bildschirmes und holt erneut die Bildschirmparameter auf
die o.g. Variablen zuriick. Es setzt auch den Text-Mode.

1000 A=100: GOTO 20: REM ### ZUTEXT ###

1010 HT=HO: VT=VE

1020 FOR X=1 TO 200

1030  GOSUB 260: HO=RV*HT: GOSUB 260: VE=RV*VT
1040 GOSUB 110: GOSUB 260: PRINT CHR$(32+RV*92);
1050 NEXT X

1060 HO=0: VE=0: GOSUB 110

1070 PRINT "O=Ende l=weiter ";

1075 GOSUB 210: Z=VAL(IN$)

1080 IF Z=1 THEN GOSUB 100: GOTO 1020

1090 GOTO 950

1100 REM ----Programm-Ende------

30000 REM Es werden Zufallszeichen an zufaellige

30010 REM Bildschirmpositionen gelegt. Beobachten
30020 REM Sie das entstehende Muster und

30030 REM eventuell sich ausbildende Texte.

32000 REM --------mmmmmmmiiia o

32010 REM H. Voelz; 9.5.89; 24.8.89 fuer Rundfunk
32020 REM XT-compatibler Rechner

Das folgende Programm "TASTE" verwendet zwei weitere BASICODE-Befehle, von
dem einer nur selten ein Aquivalent im normalen BASIC besitzt.

1) Mit GOSUB 150 wird eine vorher in SR$ abgelegte Zeichenkette auffallig
auf den Bildschirm gebracht. Bei den meisten Bascodern 1ist dies eine
inverse Zeichendarstellung. Zur besseren Hervorhebung werden an SR$ da-
bei noch rechts und links je drei Leerzeichen angefiigt. In dem Programm
wird aus dem Wert von HO nach GOSUB 100 so die Lage des auffalligen
Textes gemdl HO=INT(HO/2)-7 berechnet, dal er bei jedem Rechner in der
Mitte auf dem Bildschirm steht. Der Zahlenwert 7 ergibt sich aus der
halben Lange von SR$ plus die drei Leerzeichen.

2) Mit GOSUB 250 wird ein kurzer Ton (BEEP) erzeugt, der als Hinweis fir
besondere Ereignisse verwendet werden kann, z.B. wenn ein langer rech-
nendes Programm zum Ende gekommen ist.

Hauptsachlich dient das Programm jedoch zur genaueren Demonstration der
Eingaberoutine GOSUB 210. Sie libergibt namlich den Tastencode an zwei Vari-
ablen:

IN erhalt ASCII-Werte in leicht modifizierter Art. So werden z.B. meist nur
die Zahlenwerte fir GroBbuchstaben Ubergeben, auch wenn Kleinbuchstaben
betatigt wurden. IN$ enthdlt das Zeichen, also z.B. "A" bzw. "a" bei IN=65.
Ferner werden folgende Steuerzeichen iibergeben:

13 fir ENTER, CR 127 fur DELETE
28 fiur Cursor nach links 29 fir Cursor nach rechts
30 fur Cursor nach unten 31 fir Cursor nach oben

Es ist zu beachten, daR diese Ubergaben sehr rechnerspezifisch sein kénnen.
Deshalb ist ein Experimentieren mit dem Programm sehr nitzlich. Beachten
Sie aber beim Schreiben von BASICODE-Programmen die o.g. Einschrankungen.
Es sei noch erwahnt, daB sich die mit GOSUB 210 verwandte Routine GOSUB 200
beziglich der Variablen gleichwertig verhdlt. Hierbei wird jedoch nicht die
Tastenbetatigung abgewartet. Wenn beim Aufruf keine Taste berihrt wurde,
betragen IN=0 und IN$="".

1000 A=100: GOTO 20: REM ### TASTE ###

1010 SR$="Tastentest": HO=INT((HO-LEN(SR$)-6)/2)
1020 VE=0:GOSUB 110: GOSUB 150: PRINT

1030 PRINT"Tasteneingabe; 0 fuer Ende ": GOSUB 250
1040 GOSUB 210: PRINT IN$;IN: IF IN$<>"0" THEN 1040
1050 GOSUB 250: GOTO 950

1060 REM ----Programm-Ende------

30000 REM Test der Tastatureingabe gemaess GOSUB 210
30010 REM und Demonstration von GOSUB 150, 250
32000 REM ----------cmmmmmiiaao o

32010 REM H. Voelz; 9.5.89;24.8.89 fuer Rundfunk
32020 REM XT-compatibler Rechner

Zuweilen ist es notwendig, die Tastatur beziiglich Eingaben zu sperren, da-
mit ein Programm ohne dulere Stdérung, z.B. auch 1in Schulen oder Arbeits-
gemeinschaften, ablaufen kann. Im Gegensatz zu den meisten BASIC-Dialekten
bietet auch hier BASICODE mit GOSUB 280 eine Losung an, die das folgende
Programm demonstriert. Ausgewertet wird hierbei die Variable FR. Sie muB
vor dem Aufruf belegt werden:

FR=0 STOP/BRK-Taste ist wirksam
FR=1 STOP/BRK-Taste ist unwirksam

1000 A=100: GOTO 20: REM ### STOP ###

1010 SR$="Stoptest": HO=INT(HO/2)-6

1020 VE=0: GOSUB 110: GOSUB 150: PRINT: FR=1
1030 PRINT "Stop FR =";FR: GOSUB 280

1040 PRINT "Tasteneingabe; 0 fuer aendern ": GOSUB 250
1050 GOSUB 210: PRINT IN$; IN: IF IN$<>"0" THEN 1050
1060 PRINT"O: stop moeglich; 1: nicht; 2: Ende ";
1065 INPUT FR

1070 GOSUB 250: IF FR<2 THEN 1030

1080 GOTO 950

1090 REM ----Programm-Ende------

30000 REM Test auf Moeglichkeit der Stoptaste
30010 REM also CTRL C, BRK, Stop oder aehnlich
30020 REM erfolgt mit FR und GOSUB 270

30030 REM Wenn Sie das Programm mit BRK usw.
30040 REM unterbrechen, denken Sie bitte daran,
30050 REM mit GOTO 950 Ihren Rechner wieder in
30060 REM in den Normalzustand zu versetzen.
32000 REM -------mmmmmmem e

32010 REM H. Voelz; 9.5.89;24.8.89 fuer Rundfunk
32020 REM XT-compatibler Rechner

4. Weiterhin Text-Mode

In diesem Abschnitt werden die restlichen GOSUB-Routinen des Text-Mode
behandelt.

Das Programm "TEXTUM" liest insbhesondere Uber GOSUB 220 Zeichen, die auf
dem Bildschirm stehen, in die Variable IN zurick. Es besteht also eine ge-
wisse Verwandtschaft mit dem Befehl VPEEK im KC-BASIC. AuBerdem werden wie-
der die Besonderheiten der Routinen 200 bzw. 210 wirksam. In IN$ werden
zwar GroB- und Kleinbuchstaben zurickgemeldet, jedoch in IN nur die Werte
der GroBbuchstaben. Fir "A" und "a" gilt also IN=65. Solche Routinen sind
u.a. gunstig, wenn man menugesteuert wie im CAO0S-Mode des KC 85/3 oder /4
arbeiten will. Das Programm "TEXTUM" macht allerdings einen anderen Ge-

brauch von dieser Routine. Zunachst verwendet es den Kernteil vom Programm
"ZUTEXT". Danach werden die Zufallszeichen wieder vom Bildschirm gelesen
und dicht aneinandergefiigt. Hier wurden auch - trotz der Probleme, die sich
beim Commodore infolge seiner wuniblichen Zeichenkodierung ergeben - die
Kleinbuchstaben verwendet.

1000 A=100: GOTO 20: REM ### TEXTUM ###

1010 HT=HO: VT=VE: RESTORE

1020 FOR I=1 TO 4: READ A$: PRINT A$: NEXT I

1030 PRINT"Taste druecken ";

1035 INPUT A: GOSUB 100

1040 FOR X=1 TO 200

1050  GOSUB 260: HO=RV*HT: GOSUB 260: VE=RV*VT
1060 GOSUB 110: GOSUB 260: PRINT CHR$(32+RV*92);
1070 NEXT X

1080 HO=0: VE=0: GOSUB 110

1090 GOSUB 250: PRINT "Text zusammenfassen": BH=0:BE=1
1100 FOR VE=1 TO VT

1110 FOR HO=0 TO HT

1120 GOSUB 220: IF IN=32 THEN 1170

1130 BO=HO: HO=BH: BV=VE: VE=BE: GOSUB 110
1140 PRINT CHR$(IN);

1150 HO=BO: VE=BV: GOSUB 110

1160 BH=BH+1: IF BH=HT THEN BH=0: BE=BE+1

1170 NEXT HO

1180 NEXT VE

1190 GOSUB 250: HO=0: VE=BV+1

1200 GOSUB 110: PRINT"Ich bin fertig";

1210 GOSUB 210: GOTO 950

1220 REM ----Programm-Ende------

25000 DATA "Der Commodore verfuegt ueber einen"
25010 DATA "vom ASCII-Code abweichenden Zeichen-"
25020 DATA "satz. Daher funktioniert dieses "
25030 DATA "Programm bei ihm abweichend !"

30000 REM -------cmmmmmmiiiia o

30010 REM Es werden Zufallsbuchstaben an zufaellige

30020 REM Bildschirmpositionen gelegt. Dann werden
30030 REM sie vom Bildschirm zurueckgelesen und
30040 REM hintereinander ausgegeben. Dabei werden
30050 REM Klein- in Grossbuchstaben gewandelt.
32000 REM --------mmmmmmmiiia o

32010 REM H. Voelz; 9.5.89; 24.8.89 fuer Rundfunk
32020 REM XT-compatibler Rechner

Das Programm beginnt mit einer Besonderheit, welche die Méglichkeiten von
DATA-Zeilen demonstriert. Fir die sind in BASICODE die Zeilen von 25000 an
reserviert. 1In diesem Fall enthalten sie einen Kommentar beziglich der
Probleme beim Commodore. Die DATA-Zeilen werden mit READ A$ gelesen und
anschlieBend mit PRINT A$ angezeigt. Nach diesem Prinzip ist es z.B. auch
moglich, ein Programm sich selbst dokumentieren zu lassen.

Im normalen BASIC hat man bei Sortierprogrammen ein Problem mit den Klein-
und GroBbuchstaben. Durch ihren ASCII-Code werden sie nicht gleichwertig
geordnet. BASICODE stellt u.a. zu diesem Zweck eine Routine GOSUB 330 be-
reit. Sie verwandelt alle Kleinbuchstaben - die in SR$ enthalten sind - in
GroBbuchstaben. Dann ist ein exaktes Sortieren mdglich. Genau dies nutzt
das Programm "BUBBLE" aus.

1000 A=5000: GOTO 20: REM ### BUBBLE ###

1010 SR$="Einfacher Bubble-Sort": VE=0: M=21

1020 HO=INT((HO-LEN(SR$)-6)/2): GOSUB 110: GOSUB 150: PRINT
1030 PRINT: PRINT"bitte etwas warten"

1040 DIM A$(M),B$(M),A(M)

1050 FOR I=0 TO M

1060  GOSUB 260: N=INT(10*RV)+3: SR$=""

1070 FOR J=0 TO N

1080 GOSUB 260: IF RV<.5 THEN X=ASC("A")+RV*52
1090 IF RV>=.5 THEN X=ASC("a")+(RV-.5)*52

1100 SR$=SR$+CHR$ (X)

1116 NEXT J

1120 A(I)=I: A$(I)=SR$: GOSUB 330: B$(I)=SR$
1130 NEXT I

1140 GOSUB 250: PRINT "Dies sind die Woerter:"

1150 FOR I=0 TO M: PRINT A$(I),B$(I): NEXT I

1160 PRINT"Nun wird sortiert :";

1170 FOR I=0 TO M: F=0

1180 FOR J=0 TO M-1-I

1190 IF B$(A(J))<=B$(A(J+1)) THEN 1210

1200 B=A(J): A(J)=A(J+1): A(J+1)=B: F=1

1210 NEXT J

1220 PRINT I;: IF F=0 THEN I=M

1230 NEXT I

1240 PRINT: GOSUB 250: PRINT "sortierte Woerter"

1250 FOR I=0 TO M: PRINT A$(A(I)): NEXT I

1260 GOSUB 210: GOTO 950

30000 REM ------mmmmmm e m e e e
30010 REM Es wird der indirekte Bubble-Sort verwendet.

In den Zeilen von Zeile 1070 bis 1110 wird ein Zufallswert erzeugt. Durch
eine spezielle Erzeugung der Zufallsdaten werden hier sowohl die Probleme
des Zeichensatzes des Commodore als auch die Auswahl von nur Buchstaben
gleichermaBen gelést. Der Bereich der Zufallszahlen wird mit der Entschei-
dung IF RV<.5 geteilt und es werden getrennt grofle und kleine Buchstaben
erzeugt. Dabei wird gerade wegen des Commodore von ASC("a") bzw. ASC("A")
ausgegangen. Jedes so erzeugte Wort wird anschliefend als zweites Wort
mittels GOSUB 330 in ein Wort geandert, das nur GroBbuchstaben enthalt.
Beide Varianten werden dann fir die N=21 Worter (Zeile 1150) zur Kontrolle
angezeigt.

Das Sortieren erfolgt nur bei den Wértern mit GroRbuchstaben wund zur Ver-
meidung der garbage collection indirekt Uber die Pointer in A(J). Verwendet
wird ein verbesserter Bubble-Sort, welcher die bereits sortierten Woérter
mit FOR J=0 TO M-1-I (Zeile 1180) ausblendet und lber das Flag F entschei-
det, ob noch weiter sortiert werden muB. Die Anzahl der Sortierlaufe wird
angezeigt und anschlieBend das sortierte Feld.

Das Programm "FORMAT" arbeitet ebenfalls im Text-Mode und betrifft die
richtige Formatierung von Zahlen. (In einigen Dialekten existiert PRINT
USING.) Sie sollen auf der Variablen SR stehen und werden dann mit GOSUB
310 unter Bericksichtigung von zwei Parametern formatiert dargestellt:

CT enthalt die Anzahl der darzustellenden Ziffern und Hilfszeichen (Gesamt-
lange von SR$)
CN benennt die davon hinter dem Dezimalpunkt stehenden Ziffern.

Je nach dem Rechnertyp sind hier beachtliche Leistungen, aber auch spezifi-
sche Rundungsfehler méglich. Deshalb wurde als Ausgangswert in SR$ eine
sehr lange Ziffernfolge abgelegt. Sie wird schrittweise 14-mal durch 9
dividiert und je als formatierte Zahl wund fir den Rechner (bliche Zahl
dargestellt. So wird diese Routine einschlieBlich ihrer Grenzen demon-
striert.

1000 A=100: GOTO 20: REM ### FORMAT ###
1010 SR$="Formatieren": HO=INT(HO0/2)-9
1020 VE=0: GOSUB 110: GOSUB 150: PRINT
1030 SR$="123456789123"

1035 PRINT "Zeichenzahl = ";: INPUT CT
1040 SR=VAL(SR$)
1045 PRINT "Nachkommastellen = ";: INPUT CN

1050 PRINT: PRINT SR$;TAB(CT+5);"SR"
1060 FOR I=1 TO 14

1070  GOSUB 310: PRINT SR$; " : "; SR: SR=SR/9
1080 NEXT I

1090 PRINT "O=Ende 1=weiter ";

1095 INPUT B

1100 IF B=1 THEN 1030

1110 GOTO 950

1120 REM ----Programm-Ende------

30000 REM Zeigt die Moglichkeit der Zahlen-
30010 REM darstellung. Der String ist bewusst
30020 REM so gross gewaehlt, dass auch die
30030 REM Genauigkeit der Arithmetik und die
30040 REM Rundung getestet werden kann.

32000 REM --------mmmmmmmiiia o

32010 REM H. Voelz; 10.5.89; fuer Rundfunk
32020 REM XT-compatibler Rechner

5. Der Grafik-Mode

Der Grafik-Mode ist jene Betriebsart, bei der kleine Punkte (Pixel), Linien
Kurven usw. mit hoher Auflésung gezeichnet werden. Im Gegensatz zum BASIC
bei den KC- und einigen anderen Rechnern ist dies im BASICODE eine speziel-
le Betriebsart. Sie muB eingeschaltet werden. Dies erfolgt durch den Befehl
GOSUB 600. 1In diesem Zustand sind die meisten Befehle des Text-Mode nicht
mehr zulassig. Ganz besonders sind dies GOSUB 110, 120, 150 und 220, aber
auch die BASIC-Befehle PRINT und INPUT. Fiir Textdarstellungen ist wieder
zum Text-Mode mit GOSUB 100 zurickzugehen. Dabei wird aber automatisch der
Bildschirm geldscht. Wenn im Grafik-Mode Text zum Bildschirm gebracht wer-
den soll, ist daher ein spezieller Befehl (siehe lbernachstes Programm)
notwendig.

Im Grafik-Mode existieren fir die einzelnen Rechner sehr unterschiedliche
Pixel-Anzahlen. Sie wurden ja schon mit dem Programm DATEN fir Ihren Rech-
ner angezeigt. Damit nun ein Grafik-Programm auf allen Rechnern ein gleich-
artiges Bild erzeugt, missen die Rechnerunterschiede von BASICODE automa-

tisch bericksichtigt werden. Deshalb wurden spezielle Festlegungen fiur die
Koordinaten des Bildschirms getroffen. Seine HOhe und Breite sind auf 1
normiert. Der Punkt 0,0 liegt links oben. Der Mitte des Bildschirms ent-
spricht 0.5, 0.5. Diese Koordinatenwerte werden mit den Variablen HG und VG
Ubergeben. 1Im Text-Mode erhalten sie die ganzzahligen (integer) Schreibpo-
sitionen, im Grafik-Mode dagegen Werte von 0 bis 1, wobei die 1 nicht mehr
zugelassen ist. Das Langen-HOhen-Verhaltnis ist auf 4:3 festgelegt. Fir ein
Quadrat muB sich also z.B. HO um 0.3 und VE um 0.4 andern. Anschaulich
zeigt sich also fir den Bildschirm das folgende Bild.

0 :0 1:0 Werte fir HO : VE
o mm e e e e e eeemeeeooas +
| * | *
I I
I I
I I
| 0.5 : 0.5 |
I * I
I I
I I
I I
I I
I I
I I
R e e R +
* *
0 :1 1:1

Im folgenden Programm wird nun mit diesen Mitteln ein kleines Bild gezeich-
net. Die entsprechenden Punkt-Koordinaten sind in der DATA-Zeile 25010 ab-
gelegt. Sie werden in die Felder X(I) und Y(I) lbernommen. Mit den Daten in
Zeile 25020 wird die Reihenfolge angegeben, mit der die Punkte "angelaufen"
werden. Damit dies deutlich erkennbar ist, wurde eine Zeitverzoégerung Uber
GOSUB 450 eingefliigt. Der vorher an die Variable SD lbergebene Wert bewirkt
eine Pause von SD*0.1 Sekunden. Diese Pause kann durch Betatigen einer Ta-
ste vorzeitig beendet werden.

Mit dem Befehl GOSUB 620 wird ein Punkt entsprechend den Werten in HO und
VE gesetzt. Gleichzeitig liegt dann an dieser Stelle der unsichtbare Gra-
fik-Cursor.

Mit dem Befehl GOSUB 640 wird vom Punkt des Grafik-Cursors eine Linie zum
durch HO und VE bestimmten Punkt gezogen wund der Grafik-Cursor auf diesen
Endpunkt gesetzt.

Fir die Grafik-Befehle kann Uber CN auch ein "Farbwert" Ubergeben werden.
CN=0 bedeutet die Vordergrundfarbe (immer weiB) und CN=1 die Hintergrund-
farbe (stets schwarz). Mehr als diese beiden "Farbwerte" sind im BASICODE-3
nicht vorhanden (vgl. Abschnitt 12). Fiir echte Farbdarstellungen mull also
auf die Spezifika des jeweiligen Rechners zuriickgegriffen werden. Dies
sollte aber nur ausnahmsweise und dann in den Zeilen 20000 bis 24999 mit
ausfiuhrlichen Kommentaren erfolgen (Zeilen ab 30000).

1000 A=100: GOTO 20: REM ### GRAFIK ###
1010 DIM X(4), Y(4): RESTORE

1020 FOR I=0 TO 4:READ X(I), Y(I): NEXT I
1030 GOSUB 600: CN=0: GOSUB 250

1040 HO=X(0): VE=Y(0): GOSUB 620

1050 FOR I=0 TO 7
1060  READ X: SD=5: GOSUB 450
1070  HO=X(X): VE=Y(X): GOSUB 630

1080 NEXT I
1090 GOSUB 250: GOSUB 210: GOTO 950
25000 REM---- Data-Zeilen ------

25010 DATA .25, .75, .25, .3, .75, .3, .75, .75, .5, .1
25020 DATA 1, 2, 0, 3, 1, 4, 2, 3

30000 REM ---- Programm-Ende-----
30010 REM Es wird ein Bild gezeichnet.
32000 REM --------ommmmmiia oo

32010 REM H.Voelz; 11.5.89; fuer Rundfunk
32020 REM XT-compatibler Rechner

Ein zweites Grafik-Programm zeigt bereits etwas genauer die Leistungsfahig-
keit der Grafik-Routinen in BASICODE. Es werden ineinander geschachtelte
N-Ecke gezeichnet. Mit den Eingabeparametern kdénnen Sie versuchen, die gra-
fischen Grenzen Ihres Rechners zu erproben. Vielleicht gehen Sie dazu vom
Dreieck mit einem Dreh-Winkel von 20 Grad aus und verkleinern den Winkel
und/oder erhéhen die Eckenzahl. Bei der Wahl eines zu groflen Drehwinkels
vergroRBert sich das N-Eck und es muB ein spezieller Abbruch erfolgen. Des-
halb ist das AND in Zeile 1210 notwendig. Zur Unterstitzung Ihres Gedacht-
nisses wird am Ende jedes Bildes der alte Winkel angezeigt. Dieser und der
folgende Text beziglich der Fortsetzung des Programms sind Texteingaben,
die im Grafik-Mode mittels SR$ an die Routine GOSUB 650 Ubergeben werden.
SchlieBlich sei noch auf die Klammern im Zusammenhang mit den logischen Be-
fehlen hingewiesen. GemaR Zeilen 1040 und 1210 sind sie um die logischen
Ausdriicke im Zusammenhang mit OR, AND und NOT in BASICODE (wiederum zur
Kompabilitat bezlglich aller Rechner) notwendig. Generell sind deshalb die
logischen Operatoren nur bezlglich Vergleichen zulassig. Unter anderem wird
"wahr" namlich 1in den verschiedenen Rechnern mit +1 oder -1 dargestellt.
Verboten ist folglich in BASICODE so etwas wie 7 AND 45.

1000
1010
1020
1030
1035
1040
1050
1060
1065
1070
1080
1090
1100
1110
1120
1130
1140
1150
1160
1170
1180
1190
1200
1210
1220
1230
1240
1250

A=100: GOTO 20: REM ### N-ECK ###
HO=INT(HO/2)-6: VE=0
GOSUB 110: SR$="N-ECK": GOSUB 150: PRINT
PRINT "Anzahl der Ecken ";
INPUT N: N=INT(N)
IF (N>12) OR (N<3) THEN PRINT "Fehler": GOTO 1020
Q=6.283185/N: DIM X(N), Y(N): XX=4/3: YY=.5
R=1/3: P2=1/15
PRINT"Drehwinkel = ";: INPUT W
D=W*.0174533: CN=0: REM pi/180
S=SIN(D): C=CO0S(D): P=C0S(Q): T=SIN(Q)
P4=T/(C*T+S-P*S): X(1)=P: Y(1)=T: GOSUB 600
FOR I=1 TO N-1
P=I*Q: X(I)=COS(P): Y(I)=SIN(P)
NEXT I
X(0)=1: Y(0)=0: X(N)=1: Y(N)=0
HO=YY+R*X(0): RV=XX*R: VE=YY-RV*Y(0): GOSUB 620
FOR I=1 TO N
HO=YY+R*X(I): VE=YY-RV*Y(I): GOSUB 630
NEXT I
FOR I=0 TO N
P=X(I)*C-Y(I)*S: Y(I)=Y(I)*C+X(I)*S: X(I)=P
NEXT I
R=R*P4: IF (R>P2) AND (R<.5) THEN 1140
HO=0: VE=.1l: SR=W: GOSUB 300: SR$="Winkel ="+SR$
GOSUB 650: VE=.2: SR$="1l=weiter 0=Ende"
GOSUB 650: GOSUB 210: IF IN=49 THEN 1000
GOTO 950

30000 REM Es wird ein geschachteltes N-Eck gezeichnet
30010 REM Durch den Winkel wird die Dichte bestimmt
32000 REM -------mmmmmmmieea oo

32010 REM H. Voelz; 11.5.89; fuer Rundfunk

32020 REM XT-compatibler Rechner

6. Die Tonausgabe

Eine Tonausgabe - der Ton zur Aufmerksamkeit - wurde bereits mehrfach
verwendet. Natirlich kann BASICODE etwas mehr, obwohl auch hier wieder auf
die Grenzen der "schwachsten" Rechner Rlicksicht genommen werden mufSte. Die
Tonausgabe 1ist daher nur auf einen Kanal beschrankt. Es kann also nicht
mehrstimmig musiziert werden. Fir den einen Ton kann aber Uber drei
Parameter alles Wesentliche erreicht werden:

SP die Tonhdéhe in Halbtonabstufung mit 69 als Kammerton a (etwa 440 Hz).
SD die Tondauer in Stufen zu 0.1 Sekunden.
SV die Lautstarke mit © = Null und 15 = maximal.

Die Grenzen fur die Tonhohe hangen vom Rechnertyp ab. Aber in der Regel
sind mehrere Oktaven méglich. Dennoch ist es wegen des Klanges sinnvoll,
sich auf den mittleren Bereich zu beschranken. Aufgerufen wird die Tonaus-
gabe mit GOSUB 400. Mit dem folgenden Programm wird ein einfaches Kinder-
lied gespielt und der zugehdrige Text parallel auf dem Bildschirm ange-
zeigt.

1000 A=100: GOTO 20: REM ### LIED ###

1010 RESTORE: SR$="Alle meine Entchen": VE=0

1020 HO=INT(HO/2)-13: GOSUB 110: GOSUB 150: PRINT: PRINT
1030 FOR I=1 TO 27

1040 SV=0: SD=2: GOSUB 400

1050 Sv=15: READ A$, SP, SD

1060 IF SD<O@ THEN SD=-SD: PRINT

1070 PRINT A$: GOSUB 400: SD=1: SV=0: GOSUB 400

1080 NEXT I

1090 GOSUB 210: GOTO 950

25000 REM---- Data-Zeilen -----
25010 DATA "Al",67,3,"le",69,3," mei",71,3,"ne",72,3
25020 DATA " Ent",74,6,"chen",74,6,"schwim",76, -3

25030 DATA "men",76,3," auf",76,3," dem",76,3," See",74,9
25040 DATA "schwim",76,-3,"men",76,3

25050 DATA " auf",76,3," dem",76,3," See",74,9

25060 DATA "Koepf",72,-3,"chen",72,3," in",72,3," das",72,3
25070 DATA " Was",71,6,"ser",71,6,"Schwaenz",74,-3

25080 DATA "chen",74,3," in",74,3," die",74,3," Hoeh",67,10
30000 REM ---- Programm-Ende-----

32000 REM H.Voelz; 11.5.89; fuer Rundfunk

32010 REM XT-compatibler Rechner

Auffallig sind in diesem Programm noch die Zeilen 25060 und 25080. Sie
erreichen namlich die in BASICODE maximal zugelassene Lange von 60 Zeichen.
Bis auf das Leerzeichen nach DATA sind hier auch keine redundanten Leerzei-
chen enthalten. Einige Rechner besitzen keinen langeren Eingabepuffer und
hierauf muB beim Einlesen des BASICODE-Programms Ricksicht genommen werden.
Da Leerzeichen ohnehin von BASICODE weitgehend beim Abspeichern und beim
Einlesen entfernt werden, stehen sie hier in allen Programmen vor allem zur
besseren Ubersicht. Notwendig sind Leerzeichen nur vor GOSUB. GOTO und THEN
und dabei auch nur dann, wenn das vorangehende Zeichen kein ":" oder ")"
ist.

7. Ungang mit Daten-Files

Neben Programmen sind auch oft Daten zu speichern. Ihre Zusammenfassung
wird oft als File bezeichnet. In vielen BASIC-Dialekten erfolgt dies beziig-
lich Feldern mit den Befehlen CSAVE* bzw. CLOAD*. In anderen Fallen werden
Dateien ged6ffnet (OPEN), die Daten in der Datei abgelegt oder aus der Datei
geholt. Danach ist die Datei wieder zu schlieBen (CLOSE). Dieses Prinzip
erscheint dem Unerfahrenen oft recht kompliziert, dennoch ist es effektiv
und insbesondere durch den Umgang mit Disketten wund Festplattenspeichern
bestimmt. Hier sucht man namlich nicht nach dem Ort, wo die Daten stehen
oder abgelegt werden sollen. Das ist nur typisch fir die Kassettenrecorder,
bei denen das Zahlwerk genutzt wird. Die Buchfihrung muB man dann selber
auf einem "Zettel" (bernehmen. )

Fiir BASICODE wurde die sequentielle Datei ausgewahlt. Das Offnen der Datei
bewirkt die Bereitstellung eines entsprechenden Speicherplatzes, der durch
den Namen der Datei beschrieben wird. Deshalb muB der Name hierfiir auf der
Variablen NF$ (bergeben werden. Er darf maximal 7 Zeichen enthalten. Ferner
mul mitgeteilt werden, ob in die Datei geschrieben oder aus ihr gelesen
werden soll. Dies erfolgt durch den Wert der Variablen NF. Es bedeuten

NF gerade (6, 2, 4, 6) Lesen vom Speicher,
NF ungerade (1, 3, 5, 7) Schreiben in den Speicher.

Die verschiedenen Méglichkeiten wurden deshalb gewahlt, weil an einem Rech-
ner mehrere Speicher angeschlossen sein kdnnen, z.B. mehrere Kassettenre-
corder und/oder Diskettenlaufwerke usw. Fir die einfachen Rechner mit nur
einem Speicher, z.B. Kassettenrecorder, geniigen also die beiden Werte 0/1.

Zunachst sei die Speicherung von Daten auf einem externen Medium, z.B. auf
Kassette, betrachtet. Mit NF$ und NF=1 kann nun der Speichervorgang einge-
leitet, also die Datei erdffnet, werden. (Im folgenden Programm erfolgt
dies ab Zeile 1120.) Hierzu ist lediglich GOSUB 500 aufzurufen. Dadurch
wird der Rechner in den Mode zur Dateiarbeit versetzt. AnschlieBend werden
Daten aus dem internen RAM des Rechners auf Kassette geschrieben. Hierzu
werden die Daten als SR$ an die Routine GOSUB 560 iibergeben. Dieser Schritt

kann beliebig oft wiederholt werden. Echt gespeichert sind diese Daten aber
erst dann, wenn die Datei (das File) abgeschlossen wird. Dies erfolgt mit
dem Befehl GOSUB 580. Dadurch wird auch wieder der normale Zustand des
Rechners erreicht.

Aus dem Speicher werden die Daten in analoger Weise geholt. Fir die Routine
GOSUB 500 ist dazu lediglich NF=0 (oder 2, 4, 6) zu wahlen. Jetzt wird aber
gepruft, ob der Name NF$ im File existiert. Wenn dies der Fall ist, wird
von der Routine in IN der Wert O lUbergeben. Bei anderen Werten existiert
das File nicht. Das Lesen erfolgt mit der Routine GOSUB 540 und es werden
die Strings immer auf IN$ Ubergeben. AbschlieBend muB das File wieder mit
GOSUB 580 geschlossen werden.

Generell wird nach jeder File-Routine an IN ein Code Ubergeben, der darauf
hinweist, ob der Befehl erfolgreich, d.h. fehlerfrei, realisiert werden
konnte. Es ist daher sinnvoll, immer IN abzufragen (dies wurde zur Verein-
fachung des Programms im folgenden Beispiel weggelassen). Ist IN<>-1, so
war alles in Ordnung.

Mit dem folgenden Programm koénnen Sie nun Zeichenketten, die z.B. einen
Namen und die dazugehdrende Telefon-Nr. enthalten, abspeichern. Im Gegen-
satz zu CSAVE* und CLOAD* brauchen dabei nur die wirklich belegten Feldele-
mente gespeichert zu werden. Das ist doch sehr vorteilhaft! Die Abfrage

erfolgt spater nach dem Anfang der Zeichenkette, wobei die Zeichenzahl frei
wahlbar ist.

Sie rufen also zunachst das Programm auf und starten es. Dann geben Sie die
Namen und Nummern gemeinsam ein. Wenn Sie diese Arbeit abgeschlossen haben,
geben Sie END ein. Dadurch geht der Rechner in die Dateiarbeit dber und
verlangt den Namen fir Ihre Datei. Vielleicht wahlen Sie TELEFON. Nun wer-
den die Daten abgelegt (vorher Kassettenrecorder starten!). Wenn dies er-
folgt ist, kehrt der Rechner in den normalen Mode =zuriick. Er verlangt
"1:File erzeugen 2:File lesen", und Sie kénnen entsprechend fortfahren.
Dieser Zustand wird aber auch beim Start des Programms erreicht und so ist
es méglich, ein vorhandenes File einzulesen.

1000 A=5000: GOTO 20: REM *** FILE ***

1010 M=99: SR$="File-Arbeit": VE=0: DIM A$(M)
1020 HO=INT(HO0/2)-9: GOSUB 110: GOSUB 150: PRINT
1030 PRINT"1l:File erzeugen 2:File lesen ";
1035 INPUT B

1040 IF B=1 THEN 1070

1050 IF B=2 THEN 1170

1060 PRINT"Falsche Eingabe": GOTO 1030

1070 PRINT"Eingabe von Namen: Telefon-Nr."
1080 PRINT"Beenden mit: END"

1090 FOR I=0 TO M

1100 PRINT "Nummer: ";I+1;" ";

1110  INPUT A$(I): IF A$(I)="END" THEN N=I: I
1120 NEXT I

1130 GOSUB 250: PRINT "Speichern des Files"
1140 PRINT"Name = "

1145 INPUT NF$: NF=1: GOSUB 500

1150 FOR I=0 TO N: SR$=A$(I): GOSUB 560: NEXT I

1160 GOSUB 580: GOTO 1210

1170 PRINT"Name = ";: NF=0: INPUT NF$: GOSUB 500

1180 FOR I=0 TO M

1190  GOSUB 540: A$(I)=IN$: IF IN$="END" THEN I=M

1200 NEXT I

1210 GOSUB 580: PRINT "Das File kann genutzt werden."
1220 PRINT "gesuchter Anfang des Namens; O=Ende "

1230 INPUT A$: L=LEN(A$): N=-1: IF A$="0" THEN GOTO 950

M

1240 FOR I=0 TO M

1250  IF LEFT$(A$(I),L)=A$ THEN N=I: I=M

1260 NEXT I

1270 IF N<O@ THEN PRINT "nicht vorhanden": GOTO 1220
1280 PRINT A$(N): GOTO 1220

30000 REM ---- Programm-Ende ----

30010 REM Es wird ein Datenfile erzeugt, in welchem
30020 REM gesucht werden kann. Es kann gerettet und
30030 REM geladen werden. Als Beispiel koennen hier
30040 REM Namen und Telefon-Nr. eingegeben werden.
30050 REM Beispiel: "Meyer 27 508 35"

30060 REM Gesucht wird nach dem Anfang der Zeichenkette.
30070 REM z.B. "Mey" oder "M" usw.

32000 REM -------mmmmmmmmieea o

32010 REM H. Voelz; 11.5.89; fuer Rundfunk

32020 REM XT-compatibler Rechner

8. Erganzungen zum Bascoder

Der Bascoder ist das entscheidende Programm, welches die Méglichkeiten von
BASICODE zu realisieren gestattet. Er beriicksichtigt also die Besonderhei-
ten des jeweiligen Rechners. Deshalb miissen soviel unterschiedliche Basco-
der existieren, wie Rechnertypen fir BASICODE verwendet werden. Dieses
Programm wird mit der lblichen rechnerspezifischen Art - meist Uber einen
Kassettenrecorder - eingelesen.

Neben den verschiedenen zuvor besprochenen Routinen stellt der Bascoder vor
allem die neuen Schreib- und Lese-Routinen fir das einheitliche BASICODE-
Kassetteninterface bereit. Nur lber dieses Interface ist ja der einheitli-
che Datenaustausch méglich. Hierbei wird die 0 durch eine 1200-Hz-Welle und
die 1 durch zwei 2400-Hz-Wellen dargestellt. Die Datenibertragung erfolgt
fiur jedes Byte asynchron mit einem Start- und zwei Stoppbit. Zusatzlich
wird das achte Datenbit, also das hdchstwertige, invertiert. Dadurch stehen
fir die meisten Bytes "eigentlich" drei Stoppbits bereit. Dies erhdht we-

wesentlich die Ubertragungssicherheit und erméglicht insbesondere ein
schnelles Einsynchronisieren nach einem Fehler. Durch diese Betriebsart ist
BASICODE sehr zuverlassig. Auf Mittelwelle kénnen so Entfernungen von mehr
als tausend Kilometer Uberbrickt werden.

Die BASICODE-Programme werden im ASCII-Format, also nicht mit Token, ge-
speichert. Fir die KC-Rechner entspricht dies etwa der Speicherung mit
LIST#1 und dem Laden mit LOAD#1. Genau wie in dieser Betriebsart missen die
Programme dann zunachst in die interne Darstellung umgewandelt werden. Fir
alle in BASICODE glltigen BASIC-Befehle (siehe Tabelle 1) realisiert dies
der Bascoder. Hierfir werden u.a. Begriffe wie Ubersetzen oder translate
verwendet. Fir das Schreiben von BASICODE-Programmen auf Kassette erfolgt
dann wieder die Riickwandlung in das ASCII-Format. Fiir die Speicherung sind
weiter zwei Falle zu unterscheiden. Normalerweise sollte nur das eigentli-
che BASICODE-Programm ab Zeile 1000 gespeichert werden. Es ist aber bei ei-
nigen Bascodern auch méglich, den Bascoder kombiniert mit dem Programm zu
speichern. Dies sind dann alle BASIC-Zeilen - beginnend bei Zeile 10 bis
zum Ende des BASICODE-Programms.

Der Befehl LIST hat meist im BASICODE die gleichen Eigenschaften wie im
BASIC 1Ihres Rechners. Dies bedeutet, daR in der Regel mit LIST auch die
BASIC-Zeilen Ihres Bascoders angezeigt werden. Fir die Anzeige des
BASICODE-Programms empfiehlt sich daher ein LIST 1000-. Diese Anzeige ist
auch deshalb wichtig, damit Sie nicht eventuell versehentlich in Ihrem
Bascoder editieren.

9. Erganzungen zu den GOSUB-Routinen

Hier werden nur jene GOSUB-Routinen bericksichtigt, fir die entsprechend
den vorangegangenen Beschreibungen noch Erganzungen notwendig sind bzw.
dort nicht behandelt wurden. Eine vollstandige Zusammenstellung enthalt Ta-
belle 2 im Anhang. Zu Anfang stehen im folgenden lediglich die ausgewahlten
Zeilennummern.

20 Ist wegen GOTO eigentlich keine Subroutine. Dieser Befehl darf und muB
einzig in Zeile 1000 stehen. Die Ubergebenen Werte fir HO, VE sind
wegen des Zahlbeginns bei Null um eins kleiner als die entsprechenden
Zeilen- und Spaltenzahlen. Die Rickkehr von dieser Initialisierungs-
routine fihrt zur Zeile 1010. Wahrend die Pixelzahlen weitgehend vom
Rechnertyp abhangen, werden bei vielen Rechnern 24 Zeilen und 40 Zei-
chen je Zeile verwendet. In jedem Fall ist es giinstig, die Werte des
jeweiligen Rechners abzufragen und dann im Programm zu benutzen.

110 Hier sollten immer die Maximalwerte von HO und VE beriicksichtigt wer-
den, um eine optimale Bildschirmdarstellung zu erreichen. Beispiele
enthalten die Programme "ZUTEXT" in Zeile 1030 und "TASTE" in Zeile
1010. Die Werte von HO und VE werden durch den Aufruf nicht verandert.

150 Hierbei ist zu beachten, daB die Lange des Strings einschlieRlich der
2 mal 3 Leerzeichen nicht lber das Zeilenende hinausgeht.

200 Es sei noch einmal auf die rechnerspezifische Ausgabe in die Variablen
IN und IN$ sowie auf den Unterschied fir beide und die zulassigen Steu-
erzeichen hingewiesen. PRINT CHR$(IN) ist nicht sinnvoll.

220 Diese Routine 1liefert nur einen Wert in IN. IN$ wird durch sie also
nicht verandert. In IN werden aber nur die ASCII-Werte fir GroBbuchsta-
ben l(bergeben. Wenn auf der Position des Bildschirmes ein Kleinbuchsta-
be steht, wird der Wert in IN den des zugehdrigen GroBbuchstabens an-
nehmen (vgl. Abschnitt 12). Werden Koordinaten auBerhalb des Schirmes
angesprochen, so ist IN=0.

270

300

310

350

360
400

450

500

Sofern der Stringraum entsprechend organisiert ist, wird zunachst eine
garbage collection ausgefihrt. Dann werden die freien Speicherplatze
(Byte) berechnet. Im Gegensatz zu FRE werden der Speicherplatz von
FRE(X) und FRE(X$) addiert auf die Variable FR libergeben. Die kann z.B.
dazu benutzt werden, ein Array maximal zu dimensionieren.

Im Gegensatz zur vergleichbaren, in BASICODE nicht zugelassenen Anwei-
sung STR$ werden hier alle Leerzeichen (Spaces) entfernt.

Wie das Beispielprogramm "FORMAT" demonstriert, kann der String durch-
aus mehr Stellen als die eigentliche Zahl enthalten. Bei einigen Rech-
nern werden dadurch die Eigenschaften der implementierten Arithmetik
sichtbar. Bei der meist gebrauchlichen Binararithmetik zeigen die zu-
satzlichen Stellen dann die Konvertierungsfehler beim Ubergang von der
Dezimalzahl (im ASCII-Code) an. Diese Fehler werden sonst meist nur bei
der Subtraktion in der Umgebung von Null sichtbar.

Diese Routine wurde nicht besprochen. Sie wird nur im Zusammenhang mit
Druckern verwendet und sendet den String SR$ an den Drucker. Initiali-
sierungsroutinen fir Drucker enthalt BASICODE in der Regel nicht. Es
besteht eine gewisse Analogie zum unerlaubten BASIC-Befehl LPRINT bzw.
PRINT#2.

Sendet Newline, also 0D, OA (CRLF) an den Drucker.

Die Subroutine wird erst beendet, wenn der Ton abgelaufen ist. Es gel-
ten folgende Datengrenzen, die aber nicht von jedem Rechner voll er-
fillt werden:

SP (Sound Pitch) fur die Tonhohe, 0 < = SP < = 127
Kammerton a bei SP=69, zentrales C bei SP=60
eine ganze Zahl bedeutet einen Halbtonschritt.

SD (Sound Duration) fir die Dauer, 1 < =SD < =2

SV (Sound Volume) flur die Starke, 0 <=SV <=1
= 0 bedeutet kein horbarer Ton, Pause.

55
5

Der Abbruch dieser Routine kann auf zwei Arten erfolgen: durch Ablauf
der Zeit oder vorher durch Betatigen einer beliebigen Taste. Im ersten
Fall gilt SD=0, IN$="" und IN=0. Im zweiten Fall enthalt SD die noch
verbliebene Restzeit und IN, IN$ die Werte gemal GOSUB 200, 210. Fir
eine nicht unterbrechbare Warteroutine kann GOSUB 400 mit SV = 0 be-
nutzt werden.

Die Routinen GOSUB 500, 540, 560 und 580 verwenden teilweise die unter-
ste Bildschirmzeile fir Anweisungen und Meldungen. Sie sind nur im
Text-Mode nutzbar. Als Statusmeldung fungiert die Variable 1IN wie
folgt:

IN=0 : Fehlerfreier Verlauf.
IN=1 : Es wurde jetzt oder friher der letzte String gelesen.
IN=-1: Ein nicht behebbarer Fehler ist aufgetreten.

Mit dem Code in NF werden vorbereitet:

NF = 0 : Lesen von einer BASICODE-Kassette

NF = 1 : Schreiben auf eine BASICODE-Kassette

NF = 2 : Lesen von einem rechnerspezifischen externen Speicher

NF = 3 : Schreiben auf einen rechnerspezifischen externen Speicher
NF = 4 : Lesen von Diskette

NF = 5 : Schreiben auf Diskette

NF = 6 : Lesen von einer zweiten Diskette

NF = 7 : Schreiben auf einer zweiten Diskette

Nur fir 0 und 1 wird universelle Lesbarkeit gemaR BASICODE bewirkt. Auf
der untersten Bildschirmzeile kénnen bei der File-Arbeit nltzliche
Hinweise erscheinen. Bei der Diskettenarbeit stellen die meisten Basco-
der die Moglichkeit von bis zu drei gedffneten Files bereit.

540 Im Normalfall ist IN=0 und IN$ enthalt den gelesenen String. Bei Lese-
fehlern wird IN=-1. Der letzte String wird mit IN=1 an IN$ Ubergeben.
Bei weiteren Zugriffen bleibt IN=1 und IN$ wird ein Leerstring. Wenn
auf eine nicht gedffnete Datei zugegriffen wird, erfolgt eine Fehler-
meldung.

600 Die Subroutinen GOSUB 110, 120, 150 und 200 sind verboten. Im Grafik-
Mode ist auch keine File-Arbeit méglich. Der Grafik-Mode mufR mit GOSUB
100 beendet werden.

620 Beachten Sie die unterschiedlichen Mallstabe in X- und Y-Richtung von
4:3 und die Belegung mit dem Farbwert CN.

630 Die Linie geht von der aktuellen (unsichtbaren) Cursor-Position zu den
Pixel-Koordinaten HO, VE.

650 Der unsichtbare Cursor ist die linke obere Ecke des Schriftbeginns. Es
ist darauf zu achten, daB die Schrift nicht den rechten Bildrand uber-
schreitet. Die Schrift erscheint meist im normalen Textformat. Nach der
Schriftausgabe ist der Ort des grafischen Cursors in HO und VE enthal-
ten.

10. Hinweise zu den Variablen

Fir die Variablen gibt es im BASICODE Einschrankungen, die in den Tabellen
5 und 6 zusammengefalt sind. Variablen in BASICODE bestehen aus ein oder
zwei Zeichen, wovon das erste ein GroBbuchstabe sein mufl. Anflgen von %, !
oder # ist nicht erlaubt. Alle Variablen besitzen das Format real und je
nach Rechnertyp zumindest 6 giiltige Mantissenziffern. Stringvariablen sind
durch Anhangen von $ gekennzeichnet.

Logische Variable werden indirekt realisiert und kdénnen nur ‘wahr‘ oder
‘falsch’ sein. Mit logischen Variablen darf nicht numerisch gerechnet wer-
den, denn ‘wahr’ wird in Rechnern unterschiedlich durch +/- 1 oder andere
Werte realisiert. Deshalb ist etwa A=3*(B=1) nicht erlaubt. Statt dessen
ist beispielsweise zu verwenden A=0: IF B=1 THEN A=3.

Bevor eine Variable verwendet wird, muB ihr ein Wert zugewiesen werden.
Nicht jeder Rechner setzt die Variablen am Programmbeginn zu Null.

Fir ein schnelles Programm ist auch die Reihenfolge dieser Initialisierung
zu beachten. 0ft gebrauchte Variablen sollten zuerst - auch eventuell mit
Dummy-Werten - belegt werden.

Arrays sollten ebenfalls vor jeder Anwendung dimensioniert werden. Nicht
alle Rechner wahlen sonst automatisch den Wert 10.

11. Zu den erlaubten BASIC-Befehlen

Hier werden nur eventuelle Abweichungen vom iiblichen Gebrauch notiert. Eine
Liste der erlaubten Befehle enthalt Tabelle 1 im Anhang.

AND Es darf nur auf logische Ausdricke (Vergleiche) bezogen werden. Die
zugehdrigen Ausdricke sollten immer geklammert werden. Also z.B.
(X>45) AND (X<70). Zuweisungen von logischen Ergebnissen an eine
Variable sind zur besseren Ubersicht und Verkiirzung zulassig. Also
z.B.: Q = (A=5) AND (B=0): IF Q THEN ...

ASC

CHR$

DIM

FOR. .

GOSUB

GOTO

INPUT

LOG

MID$
NEXT
NOT
ON

OR
PRINT

REM

Nicht alle Computer liefern in allen Fallen exakt den gleichen Wert.
Das ist jedoch fir die Buchstaben weitgehend erfillt. Die Nutzung der
Funktion sollte daher mit Umsicht wund nur fir ein Zeichen erfolgen.
(Besonderheiten beim Commodore beachten!)

Bei Werten, die kleiner als 32 sind (Steuerzeichen) empfiehlt sich
Vorsicht.

hat dimmer vor der Nutzung von Feldern zu erfolgen. In BASICODE sind
maximal 2 Dimensionen, also z.B. DIM A(3,15) zugelassen. Das Feld
beginnt stets mit dem nullten Element, also A(0,0).

.TO...STEP...NEXT: Die Schleife wird mindestens einmal durchlaufen.

STEP und der Wert danach darf weggelassen werden, dann betragt die
Schrittweite 1. Nach NEXT muB immer die zugehdrige Variable stehen.
Zu einem FOR ist auch nur ein NEXT zulassig. D.h. die Schleife kann
nur an einer einzigen Stelle verlassen werden. Aus der Schleife darf
nicht herausgesprungen werden. Vorzeitiges Verlassen (iber eine Bedin-
gung erfolgt durch Setzen des Laufparameters auf seinen Endwert und
Sprung zum NEXT.

Es mul eine Zahl und keine Variable als Zeilennummer folgen. Eine
nicht existierende Zeilenzahl darf nicht verwendet werden. 1IF ...

GOSUB ist nicht zuldssig, richtig ist IF ... THEN GOSUB.

Wie GOSUB, mit Ausnahme von 20 und 950. Auch IF ... GOTO ist nicht
zulassig. Auch, wenn meist IF ... THEN Zeilennummer funktioniert,
sollte konsequenterweise IF ... THEN GOTO benutzt werden.

nur fir eine numerische oder String-Variable, Prompt nicht zulaessig,
also statt INPUT“Zahl “;Z - PRINT“Zahl “;: INPUT Z

bezieht sich auf die Basis e. Es ist zu beachten, daB in einigen
Rechnern hierfir auch LN existiert, was nicht verwendet werden darf.
In diesen Fallen kann sich LOG falschlicherweise auf den dekadischen
Logarithmus beziehen.

erfordert drei oder zwei Werte. MID$(A$,5) ist erlaubt.

verlangt die Laufvariable, siehe FOR.

verlangt Klammern, siehe AND.

Nach ON darf die Variable nur die zulassigen Werte annehmen, also von
1 bis zur Anzahl der nach GOTO bzw. GOSUB stehenden Adressen.

Siehe Bemerkungen unter AND.

Zur Formatierung existieren nur das Komma, Semikolon und TAB. Mehrere
Druckauftrage hinter PRINT missen generell durch ein Semikolon ge-
trennt werden. Es wird empfohlen, TAB oder Komma durch GOSUB 110 zu
ersetzen.

Es gibt nur REM und keine alternativen Zeichen, wie ! oder ‘. 1In
einer REM-Zeile sollte kein Doppelpunkt verwendet werden.

RESTORE existiert nur ohne Zeilenzahl.

TAB

VAL

TAB(0) ist nicht erlaubt. Vorsicht! Es gibt einige Computer, die mit
1 zu zahlen beginnen. Fir die meisten ist die erste Position 0. Sub-
routine GOSUB 110 befreit von dieser Unsicherheit.

nimmt bei nicht rein numerischen Argumenten in verschiedenen Rechnern
unterschiedliche Werte an.

12. BASICODE-3C

BASICODE - in den Niederlanden geboren und seit 1981 zur Programmibertra-
gung benutzt - wurde 1986 zur Version -3 weiterentwickelt. Seit 1989 wird
dieses System in Deutschland zur Ubertragung von Computerprogrammen in der
Rundfunksendung "REM" des Deutschlandsenders Kultur verwendet.

In diesem Jahr (1991) wurde die Version BASICODE-3C vorgestellt; méglich

wurde

dies wiederum durch das Engagement vieler Hobbyisten und der Stich-

ting BASICODE, Eindhoven.

Diese Erweiterung ist durch folgende Merkmale gekennzeichnet:

1) Darstellung von Text oder Grafik auf dem Bildschirm
in wahlweise acht Farben - nur in Verbindung mit
einem Farbmonitor/TV !

2) Herstellung eines Screendumps (Textbetrieb)

Neue Ubersetzungsprogramme bzw. Erweiterungen liegen z.Zt. fiir folgende
Computer vor:

Commodore C-64,
MSX-1 und MSX-2
Schneider CPC

Philips P 2000

Ein flr die praktische Anwendung wichtiges Merkmal ist die absolute Kompa-
tibilitat von BASICODE-3 und BASICODE-3C:

Aufgestellt: > Ablauf:
BC-3 s/w > BC-3C s/w
BC-3C Farbe > BC-3 s/w

(s/w) = schwarz/weifle Bildschirmdarstellung (monochrom)

BASICODE-3C-Programme werden, falls nur ein monochromer Monitor (TV) vor-
handen ist, mit schwarz/weillem Text bzw. Grafik dargestellt.

Programmieren mit Farbe

Wie schon in der Version -3 werden die hier notwendigen Anweisungen durch
Subroutinen ersetzt. Es sind folgende Farben zugelassen:

Code Farbe

Schwarz

Blau

Rot

Violett (Magenta)
Gruen

Hellblau (Cyan)
Gelb

Weifls

NOULUANWNRKRO

Die Code-Ziffern werden den Variablen CC(0) und CC(1) zugewiesen; als De-
fault-Einstellung gilt:

CC(0)=7 > Zeichenfarbe - Weils
CC(1)=0 > Hintergrund - Schwarz

Bei Start des Programms werden diese Werte mit der Subroutine 100 (bernom-
men und die Farben entsprechend gesetzt. Sie gelten 1im Ablauf des Pro-

gramms bis zu einer Anderung und der ndchstfolgenden GOSUB 100 Anweisung.
Die Programmzeile

CC(0)=2:CC(1)=6:G0SUB 1006
liefert nach Léschen des Schirmes rote Zeichen auf gelbem Hintergrund.

Um einen hinreichenden Kontrast der Darstellung zu erreichen, empfiehlt
sich ein Unterschied von '4' zwischen den Variablen in CC(0) und CC(1).

Der Variablenname 'CC' gilt im Sinne des BASICODE-Protokolls als verboten!

Die Subroutine 150

Die Subroutine 150 bewirkt eine inverse Bildschirmdarstellung, d.h. dem
obigen Beispiel folgend werden schwarze Zeichen auf einem hellen Hinter-
grund dargestellt. Es bestehen aber noch weitere Méglichkeiten. Ein Bei-
spiel:

CC(0)=4:CC(1)=1:SR$="TEST' :GOSUB 150

Damit wird das Wort TEST in Blau auf einem griinen Untergrund ausgegeben.
Fiir den weiteren Ablauf des Programms bleibt es bei den vor der letzten
GOSUB 100-Anweisung mit CC(0) und CC(1) eingestellten Farben. Anderungen
haben bis zum nachsten GOSUB 100 keinen EinfluB auf die normale Zeichen-
Darstellung.

Farbe im grafischen Betrieb
BASICODE-3 kennt drei Subroutinen zur grafischen Darstellung:

- GOSUB 600
Léschen des Schirms
Einleiten des grafischen Betriebes

- CN=0:GOSUB 620
CN=1:GOSUB 620
Zeichnen/L6schen eines Punktes

- CN=0:GOSUB 630
CN=1:GOSUB 630
Zeichnen/L6schen einer Linie

- CN=0:GOSUB 650
CN=1:GOSUB 650
Schreiben/Ldéschen von Text

1
c
3
4
5
[
'y
8
9

—-R—-URUU-U-V
{RETURH> DRUECKEHN

Fir das Programmieren in BASICODE-3C gilt:

- Subroutine 600 léscht den Schirm und zeigt die Farbe,
die in CC(1) codiert 1ist.

- Subroutinen 620, 630, 650:

CN=0 - Grafik/Text wird in der Farbe dargestellt, die in CC(0) festge-
legt ist.

CN=1 - Grafik/Text wird geldscht, d.h. in der Farbe dargestellt,die vor
dem letzten GOSUB 600 in CC(1) codiert war.

Screendump - Hardcopy

Die schon aus BASICODE-3 bekannte Subroutine 220 lieferte beim Auslesen
der Variablen IN mit PRINT CHR$(IN) nur GroBbuchstaben ab. Dies wurde
jetzt verbessert. Durch Hinzufligen der Variablen CN (Wert computerabhén-
gig, in den Subroutinen festgelegt) wird (ber einen Drucker das Zeichen
ausgegeben, das auf dem Bildschirm gezeigt wird:

XxxX GOSUB 220:SR$=CHR$ (IN+CN):GOSUB 350
Beim Lauf von BASICODE-3C-Programmen mit einem BASICODE-3(!)- Uberset-
zungsprogramm kénnen allerdings Kompatibilitdtsprobleme auftreten.
Als Beispiel einer Hardcopy- Routine fuer BC-3C-Programme bietet sich an:
1010 HT=HO:VT=VE: REM schirmgroesse
21000 CN=0 : REM wenn bc-3c-prog. mit

21010 FOR VE=0 TO VT : REM bc-3(!) routinen laufen
21020 SR$=""

21030 FOR H0=0 TO HT #*x% DAS SUDOKU-FPUZZLE H
21040 GOSUB 220

21050 SR$=SR$+CHR$ (IN+CN) e 155 [2 5|t
21060 NEXT HO 2a| 8| & |#
21070 GOSUB 350:GOSUB 360 |||
21080 NEXT VE 3:7 11 — §
21090 RETURN CA (R §

HHHHHHHE

| ZWEI 8EM IM 3ERFELD |#
HHHEHEHHEHHBHEHHBEHHHE

Funktionstasten

In BASICODE-3(!)- Programmen sind Uber die Routinen 200, 210 und 450 neben
den Zeichentasten (ASCII 32 ... 126) lediglich sechs Steuertasten an-
sprechbar (ASCII 13, 28 ... 31 und 127).

In der Version -3C koénnen auch die - sofern vorhandenen — Funktionstasten
betdtigt wund zur Steuerung des Programmablaufs verwendet werden. Uber die
0.a. Routinen wird

IN$ = "" und fiir
F1 : IN = -1
F2 : IN = -2
F3 : IN = -3

usw. zurickgegeben.
Es bleibt zu hoffen, daB dieser knappe Uberblick iiber die Méglichkeiten,
die die neue Version - BASICODE-3C — erdffnet, einige Hérer des 'Computer-
magazins' neugierig macht und ihnen bald die erweiterten Ubersetzungs-Pro-
gramme fiur ihre Computer zur Verfigung stehen.
Verwendete Unterlagen:
Stichting BASICODE:

Definitie BASICODE-3C (April 1991)
Toelichting BASICODE-3C (April 1991)

BASICODE-3 Bulletin no. 74 und 75
10. und 17.7.91)

Friedrich Dormeier
Berlin 39 (Sept. 1991)

Anhang

Tabelle 1. Zusammenstellung der in BASICODE giltigen BASIC-Befehle

ABS  AND ASC ATN  CHR$  COS DATA DIM EXP  FOR  GOSUB
GOTO IF INPUT INT LEFT$ LEN LET LOG MID$ NEXT NOT
ON OR  PRINT READ REM RESTORE RETURN RIGHT$ SGN  SIN  SQR
STEP TAB TAN THEN TO VAL

+ - * / ~ = < > <= >= < >

Im begrenzten Umfang ist DEF FN mdglich.

Tabelle 2. Ubersicht der GOSUB-Befehle mit Hinweis auf etwa &quivalente
Befehle in KC-BASIC

20 Programmstart, System-Reset, Variable 16schen usw. CLEAR

100 Bildschirm 16schen und Text-Modus einschalten CLS

110 Cursor auf die Position HO, VE LOCATE
120 Cursor-Position in HO, VE zurtickholen VGET, POS
150 Auffalliges Anzeigen von SR$; rechts und links 3 Spaces

200 Daten einer eventuell gedrickten Taste in IN$ und IN INKEY$
210 wie 200, jedoch mit Warten auf Tastendruck

220 Holen des Zeichens aus Schirmposition HO, VE auf IN VSET$

250 Erzeugen eines kurzen Aufmerksamkeitstones BEEP

260 Zufallsvariable in RV mit 0 <= RV < 1 RND

270 Ausflihren von garbage collection und Speicherplatz in FR FRE (X)
280 Aus- bzw. Einschalten der STOP/BRK-Taste FR=0 bzw. 1

300 SR wird ohne Space in SR$ gewandelt STR$
310 wie 300, jedoch als Zahl mit CT und CN formatiert USING
330 Alle Kleinbuchstaben in SR$ in Grolbuchstaben wandeln
350 Ubergabe von SR$ an den Drucker PRINT#2
360 1in neue Zeile mit Drucker (CRLF)
400 Erzeugung eines Tons gemaR SV, SD und SP SOUND
450 Warten von maximal SD*0.1 s auf einen Tastendruck PAUSE
500 Er6ffnen eines Files mit Namen NF$ gemaR NF OPEN
540 Aus dem File wird ein String in IN$ (bergeben LOAD*
560 SR$ wird in das File geschrieben SAVE*
580 man schlieBe den Bestand mit dem Code NF ab CLOSE
600 Rechner auf graphischen Betrieb umschalten SCREEN
620 Setzen eines Punktes in die Position HO, VE mit Farbe CN PSET
630 Zeichnen einer Linie zum Punkt HO, VE in Farbe CN LINE
650 SR$ an der Position HO, VE anzeigen (Grafik-Mode)
950 Beenden des BASICODE-Mode END

Tabelle 3. Nicht erlaubte BASIC-Befehle der KC-Rechner in BASICODE-Mode

! AT BEEP  BLOAD BORDER BYE CALL* CALL CIRCLE COLOR
CLEAR CLOAD* CLOSE CLS  CONT CSAVE* CSRLIN DEEK DOKE END

FRE  INK INKEY$ INP  INPUT# INSTR JOYST KEY  KEYLIST LINE
LINES LIST# LOAD# LOAD* LOCATE NULL OPEN OUT  PAPER PAUSE
PEEK PI POKE  POSE PRESET PRINT# PSET RND RANDOMIZE RENUMBER
RUN  SOUND SPC STOP STRING$ STR$  SWITCH TROFF TRON USR

VGET# VPEEK VPOKE WAIT WIDTH  WINDOW (LN durch LOG ersetzen)

Tabelle 4. Grundsatzlicher Zeilen-Aufbau beim BASICODE

0 — 999 Bascoder (fir jeden Computer anders)
1000 Erste Zeile des BASICODE-Programms. Sie muB folgende Form
haben:

1000 A = <Wert> : GOTO 20: REM <Programmname>

1010 -19999 Eigentliches BASICODE-Programm

20000 — 24999 Subroutinen, welche unerlaubte Befehle verwenden (Tab. 4).
Sie sollen moéglichst vermieden, zumindest aber in REM-Zeilen
genau erklart werden. Ein mdgliches Beispiel sind farbige
Bilder, welche in BASICODE-3 noch nicht moéglich sind.

25000 - 29999 Eventuell bendtigte DATA-Zeilen.

30000 - 31999 REM-Zeilen, die z.B. eine kurze Beschreibung des Programms
und Literaturhinweise enthalten.

32000 - 32767 REM-Zeilen fir den Namen und die Anschrift des Autors und
andere formale Bemerkungen.

Der Zeilenabstand sollte im Programm moglichst in 10er Abstanden gewdhlt
werden.

Tabelle 5. In BASICODE verbotene Variablen (vgl. Abschnitt 12)

1. Alle Variablen, die mit dem Buchstaben 0 beginnen.

2. AS, AT, CC, DI, EI, FN, GO, GR, IF, LN, SQ, ST, TI, TI$, TO, DI$, EI$,
SQ$.

3. PI, es enthalt aber auch nicht den Zahlenwert 3.14159.

Tabelle 6. Im Bascoder verwendete Variablen mit besonderer Bedeutung
A, CC, CT, FR, HG, HO, IN, IN$, NF, NF$, RV, SD, SP, SR, SR$, SV, VE, VG.

(vgl. Abschnitt 12)

Options:

¥T: Translate BASICODE program
¥E: LIST BASICODE program
¥C: Conwvert program to BASICODE
¥W: SAVE program in BASICODE
¥5: SAVE translated program

¥B: BackuE translation program

Bemerkungen:

Mit viel Liebe und Fleill hat den Entwurf dieses Manuskriptes
und auch die Programme Herr Jacques Haubrich (Niederlande)
durchgesehen und viele wertvolle Vorschlage unterbreitet. Hier-
fir méchten wir ihm ganz herzlich danken. Dennoch kdénnen sich
beim Editieren neue Fehler eingeschlichen haben. Deshalb gilt
dieses Material nur als Arbeitsversion. Mit der geplanten Aus-
gabe eines weitaus ausflihrlicheren Handbuches sollen diese
Fehler behoben und neue Erkenntnisse beriicksichtigt werden.
Wir bitten daher alle HOorer, uns umgehend Vorschlage, Mangel
und Fehler mitzuteilen.

